

2024年10月18日



log-rank検定と層別log-rank検定の計算過程を SASで考える

イーピーエス株式会社

臨床開発事業本部 データサイエンスセンター
統計解析 2 部
飯田 絢子

目次

1. 用語,添え字について
2. log-rank検定と層別log-rank検定の概説
3. log-rank(通常/層別)検定とCMHの関係
4. 通常のlog-rank検定の計算イメージ
5. 層別log-rank検定の計算イメージ
6. 通常のlog-rank検定の再現
7. 層別log-rank検定の再現
8. まとめ

用語，添え字について

- **用語**

- リスク人数：「number at risk」や「リスク集合」とも呼ばれるが，
本発表では「リスク人数」とする
- 通常のlog-rank：本発表において，層別log-rank検定に対して，
層別でない場合のlog-rankを指す。
- 層別(予後)因子：アウトカムに影響のある因子。「交絡因子」と記載されることもある

- **使う添え字**

- i ：層番号
- j ：時点番号

log-rank検定と層別log-rank検定の概説

log-rank検定とは

- 帰無仮説H0 : $S_1(t)=S_2(t)$. 群間で生存時間関数が同じである
- 対立仮説H1 : $S_1(t)\neq S_2(t)$. 群間で生存時間関数が異なる

$$\hat{S}(t) = \prod_{t_j < t} \left(1 - \frac{d_j}{n_j}\right)$$

d_j : 時点 t_j でのイベント発生人数

n_j : number at risk. 全体の例数 - 時点 t_j より前に死亡または打ち切りを起こした人数

※添え字 j : よく「 i 」が用いられますが、以降で層番号「 i 」を用いるため、時点を「 j 」としています

- $S_1(t)=S_2(t)$ か、どのように調べる？

死亡, 打ち切り : カテゴリカルデータ

→分割表を用いて調べる

		Outcome		合計
		反応あり	反応なし	
Factor	Drug	a	b	m1
	Placebo	c	d	m2
Total		n1	n2	N

生存時間解析以外で見る分割表

		Outcome		リスク人数
		イベント発生	イベント未発生	
Factor	Drug	O_{1j}	$N_{1j} - O_{1j}$	N_{1j}
	Placebo	O_{2j}	$N_{2j} - O_{2j}$	N_{2j} ($N_j - N_{1j}$)
Total		O_j	$N_j - O_j$	N_j

生存時間解析での分割表(j 時点)
以降, log-rankの分割表と呼ぶ.

層別解析とは

- 層別(予後)因子：アウトカムに影響のある因子。「交絡因子」と記載されることもある
- 層別解析：層別因子を考慮した解析.交絡バイアスを防ぐために層ごとに解析し, 層ごとの結果を1つの結果に統合する

層別log-rank検定における結果の統合は, どのように計算しているのか?

計算過程を後に紹介する

log-rank(通常/層別)検定とCMHの関係

通常のlog-rankおよび層別log-rank

i 番目の層におけるj 時点目 ※通常のlog-rankは層「i」なし

※j: イベントが発生した時点のみが対象。
 イベント発生最終時点はtとする(j=1~t)
 層はk 個あるとする(i=1~k)

		Outcome		リスク人数
		イベント発生	イベント未発生	
Factor	Drug	O1ij (event_d)	N1ij - O1ij	N1ij (ret_lag_risk_d)
	Placebo	O2ij (event_p)	N2ij - O2ij	N2ij (Nij-N1ij) (ret_lag_risk_p)
Total		Oij	Nij-Oij	Nij

上記表のDrug × イベント発生セル位置(赤枠で囲った位置)において,

$$E1ij = N1ij * Oij / Nij$$

$$V1ij = Oij * N1ij * (Nij - Oij) * N2ij / ((Nij * 2)(Nij - 1))$$

$$\chi^2 = Z_1^2 = \frac{[\sum_{i=1}^k \sum_{j=1}^t (O1ij - E1ij)]^2}{\sum_{i=1}^k \sum_{j=1}^t V1ij}$$

$$= \frac{[\sum_{i=1}^k \sum_{j=1}^t (O1ij - N1ij * Oij / Nij)]^2}{\sum_{i=1}^k \sum_{j=1}^t \frac{Oij * N1ij * (Nij - Oij) * N2ij}{Nij^2 (Nij - 1)}}$$

(): プログラムでの変数名

※層別log-rankのプログラム上ではは出てこない

(sex=F, Mで層別. i=1は"F", i=2は"M"のように置き換えて考える)

CMH(Cochran-Mantel-Haenszel検定)

石本 りさ. Cochran-Mantel-Haenszel検定の概要とSASでの実装方法. SASユーザー総会2024. <https://sas-user2024.ywstat.jp/download.html?n=10&key=paper>, Accessed Oct 10, 2024

交絡因子により層別化された層が K 個あると仮定した際の、i 層目の分割表:

Table 2.1 クロス集計表のサンプル

		Response		Total
		Better	Same	
Treatment	Drug	a _i	b _i	m _{1i}
	Placebo	c _i	d _i	m _{2i}
Total		n _{1i}	n _{2i}	N _i

$$E(a_i) = \frac{n_{1i} m_{1i}}{N_i}$$

$$V(a_i) = \frac{n_{1i} m_{1i} n_{2i} m_{2i}}{N_i^2 (N_i - 1)}$$

$$\chi^2_{MH} = Z^2 = \frac{[\sum_{i=1}^k (a_i - E_i)]^2}{\sum_{i=1}^k V_i} = \frac{[\sum_{i=1}^k (a_i - n_{1i} m_{1i} / N_i)]^2}{\sum_{i=1}^k \frac{n_{1i} m_{1i} n_{2i} m_{2i}}{N_i^2 (N_i - 1)}}$$

文字が違って見づらいが、分割表に対応する位置を見ると、同じ式を使っていることが分かる

※層別log-rankは時点jを考慮するので、式としては $\sum_{j=1}^t$ が追加される点のみCMHの式と異なる

通常のlog-rankの計算イメージ①

時点 ↓	d群(Drug X群)			p群(Placebo群)		
	イベント発生数 ↓	打ち切り発生数 ↓	リスク人数 ↓	イベント発生数 ↓	打ち切り発生数 ↓	リスク人数 ↓
Days	Event_d	Censr_d	ret_lag_risk_d	Event_p	Censr_p	ret_lag_risk_p
0	0	0	20	0	0	20
156	0	0	20	1	0	20
157	0	0	20	1	0	19
171	1	0	20	0	0	18
179	0	0	19	0	0	18
180	0	0	18	1	0	18
206	0	0	18	2	0	17
209	0	0	18	1	0	15
211	0	0	18	1	0	14

データから、
時点ごとの各種発生数やリスク人数の情報を作成する

このような、両群イベント0のデータ行については
分割表にまとめない

Days	Event_d	Censr_d	ret_lag_risk_d	Event_p	Censr_p	ret_lag_risk_p
0	0	0	20	0	0	20
224	0	1	17	0	0	13
268	0	0	7	0	1	4
378	0	1	1	0	0	0

イベント発生時点ごとに
分割表に落とし込む

		j=t		...		j=4 (Day 179)		j=3 (Day 171)		j=2 (Day 157)		j=1 (Day 156)	
		Outcome	リスク人数	Outcome	リスク人数	Outcome	リスク人数	Outcome	リスク人数	Outcome	リスク人数	Outcome	リスク人数
Factor	Drug	O1j (event_d)	N1j - O1j	N1j (ret_lag_risk_d)	N1j (Nj - N1j)	N2j (Nj - N1j)	Nj						
	Placebo	O2j (event_p)	N2j - O2j	N2j (ret_lag_risk_p)									
Total		Oj	Nj - Oj	Nj									


時点ごとの結果の統合は、どのように計算しているのか？

計算過程を次ページで紹介する

通常のlog-rankの計算イメージ②

イベント発生時点ごとに
分割表に落とし込む

Factor		Outcome		リスク人数
		イベント発生	イベント未発生	
		O _{1j} (event_d)	N _{1j} - O _{1j}	
Drug				N _{1j} (ret_lag_risk_d)
Placebo				N _{2j} (N _j -N _{1j}) (ret_lag_risk_p)
Total		O _j	N _j -O _j	N _j



$$E_{1ij} = N_{1ij} * O_{ij} / N_{ij}$$

$$V_{1ij} = O_{ij} * N_{1ij} * (N_{ij} - O_{ij}) * N_{2ij} / ((N_{ij} * 2) * (N_{ij} - 1))$$

$$\chi_L^2 = Z1^2 = \frac{\left[\sum_{j=1}^t (O_{1j} - E_{1j}) \right]^2}{\sum_{j=1}^t V_{1j}}$$

・ イベント発生時点ごとに $(O_{1j} - E_{1j})$ が計算される

→ t時点まで足して二乗 $\left[\sum_{j=1}^t (O_{1j} - E_{1j}) \right]^2$

・ イベント発生時点ごとに V_{1j} が計算される

→ 時点tまで足す

$$\chi_L^2 = z1^2 = \left[\sum_{j=1}^t (O_{1j} - E_{1j}) \right]^2 \div \sum_{j=1}^t V_{1j}$$

・ $prob = 1 - probchi(z1, 1)$.

probchi(確率変数の値z1,自由度) : 指定した自由度のカイ二乗分布において確率変数が統計量z1以下となる範囲の確率

層別log-rankの計算イメージ①



データを層別し、
時点ごとの各種発生数やリスク人数の情報を作成する

男性(i=1)

女性(i=2)

時点 ↓	男性層 (i=1) ↓	d群(Drug X群)			p群(Placebo群)		
		イベント 発生数 ↓	打ち切り 発生数 ↓	リスク人数 ↓	イベント 発生数 ↓	打ち切り 発生数 ↓	リスク人数 ↓
Days	Sex	Event_d	Censr_d	ret_lag_risk_d	Event_p	Censr_p	ret_lag_risk
0	M	0	0	11	0	0	9
237	M	0	0	11	1	0	9
242	M	0	0	11	1	0	8
249	M	0	0	11	1	0	7
253	M	0	0	11	1	0	6
255	M	1	0	11	0	0	5
256	M	1	0	10	0	0	5


時点 ↓	女性層 (i=2) ↓	d群(Drug X群)			p群(Placebo群)		
		イベント 発生数 ↓	打ち切り 発生数 ↓	リスク人数 ↓	イベント 発生数 ↓	打ち切り 発生数 ↓	リスク人数 ↓
Day	Sex	Event_d	Censr_d	ret_lag_risk_d	Event_p	Censr_p	ret_lag_risk_p
0	F	0	0	9	0	0	11
156	F	0	0	9	1	0	11
157	F	0	0	9	1	0	10
171	F	1	0	9	0	0	9
179	F	1	0	8	0	0	9
180	F	0	0	7	1	0	9
206	F	0	0	7	2	0	8
209	F	0	0	7	1	0	6
211	F	0	0	7	1	0	5




層別log-rankの計算イメージ②

男性(i=1)

Factor	Outcome		リスク人数
	イベント発生	イベント未発生	
Drug	O1ij (event_d)	N1ij - O1ij	N1ij (ret_lag_risk_d)
Placebo	O2ij (event_p)	N2ij - O2ij	N2ij (ret_lag_risk_p)
Total	Oij	Nij - Oij	Nij



女性(i=2)



$$\left[\sum_{j=1}^t (O1j - E1j) \right] \quad \sum_{j=1}^t V1j$$

$$\left[\sum_{j=1}^t (O1j - E1j) \right] \quad \sum_{j=1}^t V1j$$

$$\chi_L^2 = \left[\left[\sum_{j=1}^t (O1j - E1j) \right]_{\text{男性}} + \left[\sum_{j=1}^t (O1j - E1j) \right]_{\text{女性}} \right]^2 \div \left[\sum_{j=1}^t V1j \right]_{\text{男性}} + \left[\sum_{j=1}^t V1j \right]_{\text{女性}}$$

層別データに関して,

- 各時点×層別に(O1j-E1j)が計算される→層別にt時点まで足す→男性の $\sum_{j=1}^t (O1j - E1j)$ と女性の $\sum_{j=1}^t (O1j - E1j)$ を足してから二乗
- 各時点×層別にV1jが計算される→層別に時点まで足す→男性の $\sum_{j=1}^t Vj$ と女性の $\sum_{j=1}^t Vj$ を足す

$$\chi_L^2 = Z1^2 = \left[\sum_{i=1}^k \sum_{j=1}^t (O1ij - E1ij) \right]^2 \div \sum_{i=1}^k \sum_{j=1}^t V_{ij}$$

• prob = 1 - probchi(z1, 1)

本日の本題：log-rank検定と層別log-rank検定の 計算過程をSASで考える

使用データセット

[SAS Help Center: Getting Started: LIFETEST Procedure](#) の例を用いる

```
data exposed;  
  input Days  Status Treatment Sex $ @@;  
  format Treatment Rx. ;  
  datalines;  
179 1 1 F 378 0 1 M  
256 1 1 F 355 1 1 M  
262 1 1 M 319 1 1 M  
256 1 1 F 256 1 1 M  
255 1 1 M 171 1 1 F  
224 0 1 F 325 1 1 M  
225 1 1 F 325 1 1 M  
287 1 1 M 217 1 1 F  
319 1 1 M 255 1 1 F  
264 1 1 M 256 1 1 F  
237 0 0 F 291 1 0 M  
156 1 0 F 323 1 0 M  
270 1 0 M 253 1 0 M  
257 1 0 M 206 1 0 F  
242 1 0 M 206 1 0 F  
157 1 0 F 237 1 0 M  
249 1 0 M 211 1 0 F  
180 1 0 F 229 1 0 F  
226 1 0 F 234 1 0 F  
268 0 0 M 209 1 0 F  
;
```

Days : 時間(日数)を表す変数
Status : 0=打ち切り 1 =イベント
Treatment : Drug X, Placebo
Sex : F, M(層別因子)

Drug X群の割付人数 : 20人
Placebo群の割付人数 : 20人

Sex = Fにおける
Drug X群の割付人数 : 9人
Placebo群の割付人数 : 11人
sex=Mにおける
Drug X群の割付人数 : 11人
Placebo群の割付人数 : 9人

	Days	Status	Treatment	Sex
1	179	1	Drug X	F
2	378	0	Drug X	M
3	256	1	Drug X	F
4	355	1	Drug X	M
5	262	1	Drug X	M
6	319	1	Drug X	M
7	256	1	Drug X	F
8	256	1	Drug X	M
9	255	1	Drug X	M
10	171	1	Drug X	F
11	224	0	Drug X	F
12	325	1	Drug X	M
13	225	1	Drug X	F
14	325	1	Drug X	M
15	287	1	Drug X	M
16	217	1	Drug X	F
17	319	1	Drug X	M
18	255	1	Drug X	F
19	264	1	Drug X	M
20	256	1	Drug X	F
21	237	0	Placebo	F
22	291	1	Placebo	M
23	156	1	Placebo	F
24	323	1	Placebo	M
25	270	1	Placebo	M
26	253	1	Placebo	M
27	257	1	Placebo	M
28	206	1	Placebo	F
29	242	1	Placebo	M
30	206	1	Placebo	F
31	157	1	Placebo	F

通常のlog-rank検定の再現

通常のlog-rank検定をプロシジャで行うと

```
proc lifetest data=Exposed  
  time Days*Status(0);  
  strata Treatment/ test=LOGRANK;  
run;
```

層に対しての同等性の検定			
検定	カイ 2 乗値	自由度	Pr > Chi-Square
ログランク	5.6485	1	0.0175

構文 :


```
proc lifetest data=[入力データセット];  
  time [時間を表す変数] * [イベント/打ち切りを表す変数(打ち切り側の水準)];  
  strata [比較したい群変数] / test=LOGRANK;  
run;
```

例) 変数名Status
0=打ち切り
1=イベント
の場合, 「Status(0)」

この計算過程をSASで考えていく

通常のlog-rankの分割表を完成させるために目指すデータセットの形

プログラム1_通常のlogrank検定_iida.sasのデータセットWk4に該当




Days	Event_d	Censr_d	ret_lag_risk_d	Event_p	Censr_p	ret_lag_risk_p
0	0	0	20	0	0	20
156	0	0	20	1	0	20
157	0	0	20	1	0	19
171	1	0	20	0	0	18
179	1	0	19	0	0	18
180	0	0	18	1	0	18
206	0	0	18	2	0	17
209	0	0	18	1	0	15
211	0	0	18	1	0	14

Censr_dとCensr_pはlog-rankの検定の計算に直接は使用しないが、ret_lag_risk_d, ret_lag_risk_pを算出するために必要

通常のlog-rank計算のためのデータセット作成①

群×Daysごとのイベント発生数と打ち切り発生数をカウント

```
/*群×Daysごとのイベント,打ち切りの件数をカウント*/  
proc sort data=Exposed;by Treatment Days;run;  
proc freq data=Exposed(where=(status^=.)) noprint;  
  tables Status/out=wk1;  
  by Treatment Days;  
run;  
/*転置*/  
proc sort data=wk1;by Days;run;  
proc transpose data=wk1 out=wk1_2 delimiter = _;  
  id Treatment Status;  
  var COUNT;  
  by Days;  
run;  
/*変数名変換*/  
data wk1_3;  
  set wk1_2;  
  rename  
    Drug_X_0=Censr_d  
    Drug_X_1=Event_d  
    Placebo_0=Censr_p  
    Placebo_1=Event_p  
  ;  
run;
```



	Days	Event_d	Censr_d	Event_p	Censr_p
1	156	.	.	1	.
2	157	:	.	1	.
3	171	:	.	1	.

変数名の接尾語, 接頭語について.
XXXXX_d : Drug_X群情報,
XXXXX_p : Placebo群情報,
Event_X : イベント発生件数,
Censr_X : 打ち切り発生件数
※「X」は何かしら文字が入る箇所

通常のlog-rank計算のためのデータセット作成②

atrisk(リスク人数)の作成

```
/*転置してDrug_Xのatrisk及びPlaceboのatrisk列を作成*/
```

```
proc sort data=o_survival out=NDK_atrisk
```

```
  nodupkey;
```

```
  by Time atrisk;
```

```
run;
```

```
proc transpose data=NDK_atrisk out=wk1_atrisk;
```

```
  id Stratum; /*Treatment*/
```

```
  by Time ; /*Days*/
```

```
  var atrisk;
```

```
run;
```

```
/*Time=0の行のatriskを埋める(両群とも割付20人のため、20を  
入れる)*/
```

```
data wk2_atrisk;
```

```
  set wk1_atrisk;
```

```
  if Time=0 and Drug_X=. then atrisk_d=20;
```

```
  else atrisk_d=Drug_X;
```

```
  if Time=0 and Placebo=. then atrisk_p=20;
```

```
  else atrisk_p=Placebo;
```

```
  rename Time =Days;
```

```
run;
```

```
ods output SURVIVALPLOT = o_survival;
```

```
proc lifetest data=Exposed;
```

```
  time Days*Status(0);
```

```
  strata Treatment/ test=LOGRANK;
```

```
run;
```

SURVIVALPLOTによるatriskの計算再現については、
プログラム0_リスク人数の計算_iida.sasを参照

	Days	Drug_X	Placebo	atrisk_d	atrisk_p
1	0	20	.	20	20
2	156	.	20	.	20
3	157	.	19	.	19
4	171	20	.	20	.

通常のlog-rank計算のためのデータセット作成③

wk2_atrisk, wk1_3データセットのマージ

```
data wk3;  
  merge wk2_atrisk wk1_3;  
  by Days;  
run;
```

d群(Drug X群)

p群(Placebo群)

イベント発生数
↓
event_d

打ち切り発生数
↓
censr_d

リスク人数
↓
atrisk_d

イベント発生数
↓
event_p

打ち切り発生数
↓
censr_p

リスク人数
↓
atrisk_p

Days	event_d	censr_d	atrisk_d	event_p	censr_p	atrisk_p
0	.	.	20	.	.	20
156	.	.	.	1	.	20
157	.	.	.	1	.	19
171	1	.	20	.	.	.
179	1	.	19	.	.	.
180	.	.	.	1	.	18
206	.	.	.	2	.	17
209	.	.	.	1	.	15
211	.	.	.	1	.	14
217	1	.	18	.	.	.

次ページからは、欠測になっている箇所を埋めていくプログラム。例えば、Placebo群でイベントか打ち切りが発生した(例ではイベントのみ発生)が、Drug_X群では何も発生していないDay156~157については、SURVIVALPLOT (o_survivalデータセット)ではリスク人数の計算がされていないため、手動でリスク人数を算出していく。Placebo群も同様に欠測しているリスク人数箇所を計算。

通常のlog-rank計算のためのデータセット作成④

リスク人数の計算

```
data Wk4;
  set Wk3;
  /*各群のイベントと打ち切りの変数について、欠測値は0件に直す*/
  if event_d=. then event_d=0;
  if Censr_d=. then Censr_d=0;
  if event_p=. then event_p=0;
  if Censr_p=. then Censr_p=0;
  /* == Drug_X群のリスク人数(ret_lag_risk_d)の計算 == */
  /*①各時点における、atrisk - イベント件数 - 打ち切り件数を計算*/
  if atrisk_d^=. then risk_d=atrisk_d - Event_d - Censr_d;
  /*②前の行の①の値を引き継ぐ*/
  lag_risk_d=lag(risk_d);
  /*③②の操作でもまだlag_risk_dが欠測の場合(欠測レコードが2連続以上)は、欠測している時点より前の、値を持つ最終行のlag_risk_dをretainする*/
  retain ret_lag_risk_d;
  if lag_risk_d^=. then ret_lag_risk_d=lag_risk_d;
  /*④1行目データ(Day 0)のret_lag_risk_dは割付人数(20人)にする*/
  if _N_=1 then ret_lag_risk_d=20;
run;
```

同様に、Placebo群のリスク人数(ret_lag_risk_p)を計算

詳細はプログラム1_通常のlogrank検定_iida.sasの
Wk4参照

リスク人数の求め方とリスク人数の欠測が続く場合の埋め方

VIEWTABLE: Work.Wk3							
	Days	event_d	Censr_d	atrisk_d	event_p	Censr_p	atrisk_p
1	0	.	.	20	.	.	20
2	156	.	.	.	1	.	20
3	157	.	.	.	1	.	19
4	171	1	.	20	.	.	.
5	179	1	.	19	.	.	.
6	180	.	.	.	1	.	18
7	206	.	.	.	2	.	17
8	209	.	.	.	1	.	15
9	211	.	.	.	1	.	14
10	217	1	.	18	.	.	.

VIEWTABLE: Work.Wk4							
	Days	Event_d	Censr_d	ret_lag_risk_d	Event_p	Censr_p	ret_lag_risk_p
1	0	0	0	20	0	0	20
2	156	0	0	20	1	0	20
3	157	0	0	20	1	0	19
4	171	1	0	20	0	0	18
5	179	1	0	19	0	0	18
6	180	0	0	18	1	0	18
7	206	0	0	18	2	0	17
8	209	0	0	18	1	0	15
9	211	0	0	18	1	0	14
10	217	1	0	18	0	0	13
11	224	0	1	17	0	0	12

Day 156
のリスク
人数は
20-0=20

Day 179の
リスク人数
は
19-1=18
(20-1-1
=18)

- ・リスク人数の計算：

リスク人数 = 割付人数 - (リスク人数を求めたい行の**前の行までの**イベントと打ち切り累計数)
 →前の行にリスク人数があれば、リスク人数がある行の、
 「リスク人数 - 同行のイベント数 - 同行の打ち切り数」が**次の行の**リスク人数となる

- ・リスク人数の欠測レコードが連続する場合、前の欠測行を埋めてからリスク人数を順に計算していく
 →プログラムが少し複雑？

通常のlog-rank計算のためのデータセット作成⑤

log-rank計算用にrename

```
data Wk5;
  set Wk4;
  rename
    ret_lag_risk_d=N1j
    ret_lag_risk_p=N2j
    event_d=O1j
    event_p=O2j
  ;
run;
```

Work.Wk4						
Days	Event_d	Censr_d	ret_lag_risk_d	Event_p	Censr_p	ret_lag_risk_p
0	0	0	20	0	0	20
156	0	0	20	1	0	20
157	0	0	20	1	0	19
171	1	0	20	0	0	18
179	1	0	19	0	0	18
180	0	0	18	1	0	18
206	0	0	18	2	0	17
209	0	0	18	1	0	15
211	0	0	18	1	0	14
217	1	0	18	0	0	13
224	0	1	17	0	0	12

時点 j の分割表

		Outcome		リスク人数 (死亡する可能性のある人数)
		イベント発生	イベント未発生	
Factor	Drug	O1j (event_d)	N1j - O1j	N1j (ret_lag_risk_d)
	Placebo	O2j (event_p)	N2j - O2j	N2j (Nj-N1j) (ret_lag_risk_p)
Total		Oj	Nj-Oj	Nj

通常のlog-rank検定の値(統計量,p値) の算出①

```
data Wk6;
  set Wk5 END=EOF;
  /*少なくとも一方の群でイベントが発生したデータに絞る*/
  where Days ^= 0 and ^(O1j=0 and O2j=0);
```

```
  Nj=N1j+N2j;
  Oj=O1j+O2j;
  E1j=N1j*Oj/Nj;
```

```
/* == z1^2の分母 == */
/*各時点*/
V1j=divide (Oj*N1j*(Nj-Oj)*(Nj-N1j),(Nj**2)*(Nj-1));
```

```
/*V1jの時点合*/
V1+V1j;
```

```
/*z1^2の分母*/
if EOF=1 then do;
  denom=V1;
end;
```

次ページに続く

時点 j の分割表

		Outcome		リスク人数 (死亡する可能性のある人数)
		イベント発生	イベント未発生	
Factor	Drug	O1j (event_d)	N1j - O1j	N1j (ret_lag_risk_d)
	Placebo	O2j (event_p)	N2j - O2j	N2j (Nj-N1j) (ret_lag_risk_p)
Total		Oj	Nj-Oj	Nj

$$E1j = N1j * Oj / Nj$$

$$V1j = \frac{Oj * N1j * (Nj - Oj) * (Nj - N1j)}{Nj^2 * (Nj - 1)}$$

↓ 時点合

$$V1 = \sum_{j=1}^t V1j$$

$$\chi_L^2 = Z1^2 = \left[\sum_{j=1}^t (O1j - E1j) \right]^2 \div \sum_{j=1}^t V1j$$

通常のlog-rank検定の値(統計量,p値) の算出②

/*data Wk6の続き*/

/* == z1²の分子 == */

/*各時間における, 期待値と実測値の差*/

$$OE1j = (O1j - E1j);$$

/*全時間における, 期待値と実測値の差(全時点和)*/

$$OE1 + OE1j;$$

/*z1²の分子*/

if EOF=1 then do;

Num=OE1**2;

end;

/*z1²の計算 ※プログラム上は変数名「z1²」が書けない為「Z1」と表記*/

if EOF=1 then do;

z1=Num/denom;

end;

時点 j の分割表

		Outcome		リスク人数 (死亡する可能性のある人数)
		イベント発生	イベント未発生	
Factor	Drug	O1j (event_d)	N1j - O1j	N1j (ret_lag_risk_d)
	Placebo	O2j (event_p)	N2j - O2j	N2j (Nj-N1j) (ret_lag_risk_p)
Total		Oj	Nj-Oj	Nj

$$E1j = N1j * Oj / Nj$$

$$OE1j = (O1j - E1j)$$

時点
和

$$OE1 = \sum_{j=1}^t OE1j$$

$$\chi_L^2 = Z1^2 = \left[\sum_{j=1}^t (O1j - E1j) \right]^2 \div \sum_{j=1}^t v1j$$

$$\text{Num} = OE1 ** 2$$

$$\text{denom} = V$$

1
(前ページ)

$$z1^2 = \text{Num} / \text{denom}$$

通常のlog-rank検定の値(統計量,p値) の算出③


```
/*data Wk6の続き*/
```

```
/*p値計算*/
```

```
if EOF=1 then do;  
  prob = 1 - probchi(z1, 1);  
end;
```

```
run;
```

- 2×2分割表の為、カイ二乗分布の自由度は1
- probchi(z1, 自由度) :
指定した自由度のカイ二乗分布において、横軸(確率変数)が統計量z1以下となる範囲の確率



z1	prob
5.6484924939	0.0174703848

層に対しての同等性の検定			
検定	カイ 2 乗値	自由度	Pr > Chi-Square
ログランク	5.6485	1	0.0175

一致.

算出した「z1」=SAS出力結果「カイ二乗値」

算出した「prob」=SAS出力結果「Pr>Chi-Square」

層別log-rank検定の再現

層別log-rank検定をプロシジャで行うと

```
proc lifetest data=Exposed ;  
  time Days*Status(0);  
  strata Sex/ group=Treatment  
  test=LOGRANK  
  ;  
run;
```

グループに対する同等性の層別検定			
検定	カイ 2 乗値	自由度	Pr > Chi-Square
ログランク	7.2466	1	0.0071

構文 :


```
proc lifetest data=[入力データセット];  
  time [時間を表す変数] * [イベント/打ち切りを表す変数(打ち切り側の水準)];  
  strata [層別変数] / group=[比較したい群変数] test=LOGRANK;  
run;
```

例) 変数名Status
0=打ち切り
1=イベント
の場合, 「Status(0)」

この計算過程をSASで考えていく

層別log-rankの分割表を完成させるために目指すデータセットの形

プログラム2_層別logrank検定_iida.sasのデータセットWk5_0に該当



VIEWTABLE: Work.Wk5_0

	SEX	Days	Event_d	Censr_d	ret_lag_risk_d	Event_p	Censr_p	ret_lag_risk_p
10	F	217	1	0	7	0	0	4
11	F	224	0	1	6	0	0	4
12	F	225	1	0	5	0	0	4
13	F	226	0	0	4	1	0	4
14	F	229	0	0	4	1	0	3
15	F	234	0	0	4	1	0	2
16	F	237	0	0	4	0	1	1
17	F	255	1	0	4	0	0	0
18	F	256	3	0	3	0	0	0
19	M	0	0	0	11	0	0	9
20	M	237	0	0	11	1	0	9
21	M	242	0	0	11	1	0	8
22	M	249	0	0	11	1	0	7
23	M	253	0	0	11	1	0	6
24	M	255	1	0	11	0	0	5
25	M	255	1	0	10	0	0	5

Censr_dとCensr_pはlog-rankの検定の計算に直接は使用しないが、ret_lag_risk_d, ret_lag_risk_pを算出するために必要

層別log-rank計算のためのデータセット作成①②

- 通常のlog-rank検定のプログラムとほぼ同じの為、省略。

プログラム2_層別logrank検定_iida.sasにおける「data wk2_atrisk;」まで省略

- 通常のlog-rank検定のプログラムと異なる箇所：層別因子であるSEXを入れる例)

```
/*性別×群×Daysごとのイベント発生数と打ち切り発生数をカウント*/
```

```
proc sort data=Exposed;by Sex Days;run;
```

```
proc freq data=Exposed(where=(status^=.) ) noprint;
```

```
  tables Treatment*Status/out=wk1;
```

```
  by Sex Days;
```

```
run;
```

層別log-rank計算のためのデータセット作成③

wk2_atrisk, wk1_3データセットのマージ

```
data wk3;  
format Sex Days event_d Censr_d atrisk_d event_p Censr_p atrisk_p;  
keep Sex Days event_d Censr_d atrisk_d event_p Censr_p atrisk_p;  
merge wk2_atrisk wk1_3;  
by Sex Days;  
run;
```

	Sex	Days	event_d	Censr_d	atrisk_d	event_p	Censr_p	atrisk_p
10	F	217	1	.	7	.	.	.
11	F	224	.	1	6	.	.	.
12	F	225	1	.	5	.	.	.
13	F	226	.	.	.	1	.	4
14	F	229	.	.	.	1	.	3
15	F	234	.	.	.	1	.	2
16	F	237	1	1
17	F	255	1	.	4	.	.	.
18	F	256	3	.	3	.	.	.
19	M	0	.	.	11	.	.	9
20	M	237	.	.	.	1	.	9
21	M	242	.	.	.	1	.	8
22	M	249	.	.	.	1	.	7
23	M	253	.	.	.	1	.	6
24	M	255	1	.	11	.	.	.

次ページからは、性別ごとにデータセットを分割した上で、欠測になっている箇所を埋めていくプログラム。
女性のPlacebo群でイベントか打ち切りが発生したが、
女性のDrug_X群では何も発生していないDay226~237については、SURVIVALPLOT (o_survival
データセット)ではリスク人数の計算がされていないため、
手動でリスク人数を算出していく。Placebo群も同様に欠測しているリスク人数箇所を計算。男性についても同様に計算

層別log-rank計算のためのデータセット作成④

リスク人数の計算

「log-rank計算のためのデータセット作成④」
スライドと同様の為、
以降の記載は省略する

```
%macro Wk4(Sex);  
data Wk4_&Sex.;  
  set Wk3;  
  /*層(性別)別データセットにしたうえで以下計算していく*/  
  where Sex="&Sex.";  
  /*各群のイベントと打ち切りの変数について、欠測値は0件に直す*/  
  /* ==性別ごとのDrug_X群のリスク人数(ret_lag_risk_d)の計算 == */  
  /*①各行のリスク人数(ret_lag_risk_d)を計算*/  
  /*②前の行のリスク人数を引き継ぐ*/  
  /*③lag_risk_dが欠測の場合は欠測前の行のlag_risk_dをretainする*/  
  /*④1行目データ(Days=0)のret_lag_risk_dは性別 * 群の割付人数にする*/  
  if Sex="F" and Days=0 then do;  
    ret_lag_risk_d=9;  
  end;  
  
  if Sex="M" and Days=0 then do;  
    ret_lag_risk_d=11;  
  end;  
  
  同様に、Placebo群のリスク人数(ret_lag_risk_p)を計算  
%mend Wk4;  
  
%Wk4(Sex=F);  
%Wk4(Sex=M);
```

BLE: Work.A (度数とパーセント)		
Sex	Treatment	COUNT
F	Placebo	11
F	Drug X	9
M	Placebo	9
M	Drug X	11

詳細はプログラム2_層別logrank検定_iida.sasの%macro Wk4参照

層別log-rank計算のためのデータセット作成⑤

log-rank計算用にrename

```
data Wk5_0;  
  set Wk4_F Wk4_M;  
  ;  
run;  
data Wk5_1;  
  set Wk5_0;  
  rename  
  ret_lag_risk_d=N1j  
  ret_lag_risk_p=N2j  
  event_d=O1j  
  event_p=O2j  
  ;  
run;
```

i 番目の層におけるj 時点目

		Outcome		リスク人数
		イベント発生	イベント未発生	
Factor	Drug	O1ij (event_d)	N1ij - O1ij	N1ij (ret_lag_risk_d)
	Placebo	O2ij (event_p)	N2ij - O2ij	N2ij (Nij-N1ij) (ret_lag_risk_p)
Total		Oij	Nij-Oj	Nij

以降スライドにおける,
プログラムと分割表の添え字記号の差について

data Wk5,%macro Wk4(Sex)において, 層ごとでデータセット内の変数名を変えていないため,
プログラムにおける変数名に「i」はない.
プログラム変数名は分割表の「i」を除いた名前と対応している.
例) プログラムの変数N1jは分割表のN1ijに対応

層別log-rank検定の値(統計量,p値) の算出①

%macro Wk6:層(Sex)ごとに統計量の要素を計算するマクロ

```

%macro Wk6(Sex);
  data Wk6_&Sex.;
  set Wk5_1 END=EOF;
  /*層別データセットに関し,
  少なくとも一方の群でイベントが発生したデータに絞る*/
  where Sex="&Sex." and Days^=0 and ^(O1j=0 and O2j=0);
  Nj=N1j+N2j;
  Oj=O1j+O2j;


  /* == z1^2の分母の要素(統合前の各層のz1^2の分母) == */
  V1j=divide (Oj*N1j*(Nj-Oj)*(Nj-N1j),(Nj**2)*(Nj-1));/*各時点*/
  V1+V1j;/*時点和*/
  if EOF=1 then do;
    denom=V1;
  end;

```

次ページに続く

i 番目の層におけるj 時点目

		Outcome		リスク人数
		イベント発生	イベント未発生	
Factor	Drug	O1ij (event_d)	N1ij - O1ij	N1ij (ret_lag_risk_d)
	Placebo	O2ij (event_p)	N2ij - O2ij	N2ij (Nij-N1ij) (ret_lag_risk_p)
Total		Oij	Nij-Oj	Nij



層別log-rank検定の値(統計量,p値) の算出②

```
/*%macro Wk6の続き*/
```

```
/* == z1^2の分子の要素(統合前の各層のz1^2の分子) == */
```

```
E1j=N1j*Oj/Nj;
```

```
/*各時間における, 期待値と実測値の差*/
```

```
OE1j=(O1j-E1j);
```

```
/*全時間における, 期待値と実測値の差(全時点和)*/
```


```
OE1+OE1j;
```

```
if EOF=1 then do;
```

```
Num=OE1; /*この時点では二乗しない*/
```

```
end;
```

```
run;
```



```
/*後のmergeのために, 変数名をrenameし, 分子と分母(最終行)のみ抽出*/
```

```
data Wk6_&Sex._2;
```

```
set Wk6_&Sex. END=EOF;
```

```
keep Days Num Denom BEF_sum;
```

```
rename
```

```
Num=Num_&Sex.
```

```
Denom=Denom_&Sex.
```

```
;
```


```
if EOF=1 then do;BEF_sum=1; output;end;
```

```
run;
```

```
%mend Wk6;
```

```
%Wk6(Sex=F); /*女性側のデータセット作成*/
```

```
%Wk6(Sex=M); /*男性側のデータセット作成*/
```



層別log-rank検定の値(統計量,p値) の算出③

```
/*層別で計算したz12の分子,分母要素についてmerge*/
```

```
data Wk7;
  merge Wk6_F_2 Wk6_M_2;
  by BEF_sum;
run;
```


```
/*層別で計算したz12の分子,分母要素について計算結果の統合*/
```

```
data Wk8;
  set Wk7;
  total_Num=Num_F + Num_M;
  total_denom=Denom_F + Denom_M;
```

```
/*z12の計算 ※プログラム上は変数名「z12」が書けない為「Z1」と表記*/
```

```
z1=total_Num**2/total_denom;
/*p値の計算*/
prob = 1 - probchi(z1, 1);
run;
```

一致.
算出した「z1」=SAS出力結果「カイ二乗値」
算出した「prob」=SAS出力結果「Pr>Chi-Square」



z1	prob
7.2465618779	0.0071036897

グループに対する同等性の層別検定			
検定	カイ 2 乗値	自由度	Pr > Chi-Square
ログランク	7.2466	1	0.0071

まとめ

- log-rank検定は，生存時間関数が群間で異なるか検定する手法。
- 生存時間関数が群間で異なるかは，以下の分割表を用いて 統計量を計算
→カイ二乗分布でp値を計算

$$\chi^2 = Z_1^2 = \frac{\left[\sum_{i=1}^k \sum_{j=1}^t (O_{1ij} - E_{1ij}) \right]^2}{\sum_{i=1}^k \sum_{j=1}^t V_{1ij}}$$

$$= \frac{\left[\sum_{i=1}^k \sum_{j=1}^t (O_{1ij} - N_{1ij} * O_{ij} / N_{ij}) \right]^2}{\sum_{i=1}^k \sum_{j=1}^t \frac{O_{ij} * N_{1ij} * (N_{ij} - O_{ij}) * N_{2ij}}{N_{ij}^2 * (N_{ij} - 1)}}$$

$$E_{1ij} = N_{1ij} * O_{ij} / N_{ij}$$

$$V_{1ij} = O_{ij} * N_{1ij} * (N_{ij} - O_{ij}) * N_{2ij} / ((N_{ij} * 2) * (N_{ij} - 1))$$

i 番目の層におけるj 時点目

		Outcome		リスク人数
		イベント発生	イベント未発生	
Factor	Drug	O _{1ij} (event_d)	N _{1ij} - O _{1ij}	N _{1ij} (ret_lag_risk_d)
	Placebo	O _{2ij} (event_p)	N _{2ij} - O _{2ij}	N _{2ij} (N _{ij} - N _{1ij}) (ret_lag_risk_p)
Total		O _{ij}	N _{ij} - O _{ij}	N _{ij}

- SASプロシジャではproc lifetest で数行で検定結果が出力できるが，上記の計算式を用いたSASプログラムで結果を再現した
- 計算のポイント：時点和や層別の結果をいつ統合するか
- 統計量Z₁²の分子と分母は別々に計算して時点和を求めてから計算する。
- 層別の場合は，各層の分子要素の時点和結果を足し，各層の分母要素の時点和結果を足したうえで統計量Z₁²を計算

参考文献

[1]SAS Help Center: Getting Started: LIFETEST Procedure

https://go.documentation.sas.com/doc/ja/pgmsascdc/9.4_3.4/statug/statug_lifetest_gettingstarted.htm, Accessed Oct 10, 2024

[2]中澤 秀夫. ー基礎科学から医学・医療を見るー生存時間データ解析と比例ハザードモデル. 日医大医学会誌 2015; 11 (1)

https://www.jstage.jst.go.jp/article/manms/11/1/11_29/_pdf, Accessed Oct 10, 2024

[3]リハビリテーション統計学_ログランク検定

<https://y2pt.com/13847/>, Accessed Oct 10, 2024

[4]折村 奈美. データステップを用いた統計検定の再現. SASユーザー総会2024. (2024)

<https://sas-user2024.ywstat.jp/download.html?n=63&key=paper>, Accessed Oct 10, 2024

[5]石本 りさ. Cochran-Mantel-Haenszel検定の概要とSASでの実装方法. SASユーザー総会2024.

<https://sas-user2024.ywstat.jp/download.html?n=10&key=paper>, Accessed Oct 10, 2024

[6]層別解析とサブグループ解析の違いは？ 統合方法や結果の見方もわかりやすく！ | いちばんやさしい、医療統計

<https://best-biostatistics.com/review/sub.html>, Accessed Oct 10, 2024

[7]サブグループ解析の解釈 | 医療統計 | ビーリンサイト

https://www.blincyto.jp/documents/medical_statistics09, Accessed Oct 10, 2024



イーピーエス株式会社